

S. GRZEGÓRSKI (Lublin)

EVALUATION OF PROBABILITY
FOR THE F -SNEDECOR DISTRIBUTION

1. Function declaration.

```
real procedure PFSnedecor(n, m, FO, error);
  value n, m, FO;
  integer n, m;
  real FO;
  label error;
  comment PFSnedecor calculates the probability  $P(F > FO)$ ,
    where FO is a given value of the  $F$ -Snedecor function.
  Data:
    FO — value of  $F$ -Snedecor test function,
    n — number of degrees of freedom for the first random
      variable (in the numerator of the  $F$ -formula),
    m — number of degrees of freedom for the second random
      variable (in the denominator of the  $F$ -formula).
  Other parameters:
    error — label indicating error exit ( $n < 1$  or  $m < 1$  or  $FO < 0$ );
  begin
    integer i, k;
    real a, PF, RF;
    Boolean B;
    switch E: = pp, pn, np, np;
    if  $n < 1 \vee m < 1 \vee FO < .0$  then go to error;
    B: = true;
    PF: = .0;
    go to if  $1.0 + FO = 1.0$  then FIN
      else  $E[2 \times (n - n \div 2 \times 2 - m \div 2) + m + 1]$ ;
  pp: if  $m \leq n$  then go to np;
  pn: k: = n;
    n: = m;
    m: = k;
```

```

    FO: = 1.0/FO;
    B: = false;
np: FO: = m/(m + n × FO);
    if m = 1 then go to mone;
    i: = m ÷ 2;
    k: = i + i;
    a: = RF: = PF: = (1.0 - FO) ↑ (n/k);
    n: = n - 2;
    m: = m ÷ 2;
    for k: = m - k + 4 step 2 until m do
        begin
Lnd:    if RF > 2.0 then
            begin
                PF: = PF × a;
                RF: = RF × a;
                i: = i - 1;
                go to Lnd
            end RF > 2.0;
            RF: = (n + k) × FO × RF/k;
            PF: = PF + RF
        end k;
    PF: = PF × a ↑ (i - 1);
    if m ÷ 2 × 2 = m then go to FIN;
    n: = n + 2;
    PF: = n × PF × sqrt(FO);
mone: FO: = 1.0 - FO;
    RF: = .0;
    a: = sqrt(FO - FO × FO);
    for k: = n step -2 until 3 do
        begin
            RF: = (FO - FO/k) × RF + a;
            PF: = PF - PF/k
        end k;
    PF: = .5 - (2.0 × (RF - PF) - arctan((FO - .5)/a)) × .31830988618;
    comment .31830988618379... = 1.0/π;
FIN: if PF > 1.0 then PF: = 1.0;
    PFSnedecor: = if B then 1.0 - PF else PF
end PFSnedecor

```

2. Method used. The well-known formula for the probability of the F -Snedecor distribution has the form

$$P(F > F_0) = n^{n/2} m^{m/2} \int_{F_0}^{\infty} F^{n/2-1} (nF + m)^{-(n+m)/2} dF / B(n/2, m/2),$$

where

$$B(p, q) = \int_0^1 t^{p-1}(1-t)^{q-1} dt.$$

Substituting $x = nF/(nF + m)$, $x_0 = nF_0/(nF_0 + m)$, and writing

$$Q(n, m, x) = \int_0^x t^{n/2-1}(1-t)^{m/2-1} dt,$$

we have $B(n/2, m/2) = Q(n, m, 1)$ and

$$(1) \quad P(F > F_0) = 1 - Q(n, m, x_0)/Q(n, m, 1).$$

Let us write

$$P(n, m, x) = x^{n/2} \sum_{k=0}^{m/2-1} \frac{(n+2k-2)!!}{(n-2k)!!(2k)!!} (1-x)^k,$$

$$R(n, m, x) = 2\sqrt{x(1-x)} \sum_{k=0}^{n/2-1} \frac{(2k)!!}{(2k+1)!!} x^k + \arctan \frac{0.5-x}{\sqrt{x(1-x)}}$$

and

$$H(n, m, x) = \begin{cases} P(n, m, x) & \text{for } m \text{ even,} \\ 1 - P(m, n, 1-x) & \text{for } n \text{ even,} \\ 0.5 + [R(n, m, x) - 2 \frac{(n-1)!!}{(n-2)!!} \sqrt{1-x} P(n, m, x)]/\pi & \text{for } n, m \text{ odd.} \end{cases}$$

With the recurrent formulas

$$Q(n, 2, x) = 2x^{n/2}/n \quad \text{for } n = 1, 2, \dots,$$

$$Q(1, 1, x) = \arctan \frac{0.5-x}{\sqrt{x(1-x)}},$$

$$Q(k, l, x) = \begin{cases} [x^{k/2-1}(1-x)^{l/2} + (k/2-1)Q(k-2, l, x)]/(k/2+l/2-1) & \text{for } k = 3, 4, \dots, \\ [x^{k/2}(1-x)^{l/2-1} + (l/2-1)Q(k, l-2, x)]/(k/2+l/2-1) & \text{for } l = 3, 4, \dots \end{cases}$$

we may prove that

$$(2) \quad Q(n, m, x) = Q(n, m, 1)^* H(n, m, x).$$

Finally, from (1) and (2) it follows that

$$P(F > F_0) = 1 - H(n, m, x_0).$$

3. Certification. *PFSnedecor* calculates the same value as *Ftest* [3] or *Fisher* [1]. All three procedures were tested and compared on the Odra 1013 computer with 31-bit floating-point mantissa, in the FALA-69 autocode. It appears that

1° *PFSnedecor* is much faster than *Fisher* and in some cases faster than *Ftest*.

2° When one of the degrees of freedom is greater than 120, *Fisher* gives floating-point overflow. When one of the degrees of freedom is greater than 128, *Ftest* uses some approximative formulae, giving a lesser accuracy. *PFSnedecor* uses exact formulae for all n and m .

3° *Ftest* is 2 times longer than *PFSnedecor*, uses 2 times more variables, and must be complemented by a procedure *Gauss* [2]. Moreover, *Ftest* exploits the functions *sin* and *cos*.

References

- [1] E. Dorrer, *Algorithm 322: F-Distribution* Comm. ACM 11 (1968), p. 116-117.
- [2] D. Ibbetson, *Algorithm 209: Gauss*, ibidem 6 (1963), p. 616.
- [3] J. Morris, *Algorithm 346: Ftest*, ibidem 3 (1969), p. 184.
- [4] W. Oktaba, *Metody statystyki matematycznej w doświadczalnictwie*, Warszawa 1967.
- [5] W. Sadowski, *Statystyka matematyczna*, Warszawa 1965.
- [6] G. W. Snedecor, *Statistical methods*, Iowa State Univ. Press, Ames, Iowa, 1956.

INSTITUTE OF MATHEMATICS
MARIA CURIE-SKŁODOWSKA UNIVERSITY
LUBLIN

*Received on 4. 11. 1970 ;
revised version on 8. 3. 1971*

S. GRZEGÓRSKI (Lublin)

ALGORYTM 15

OBLICZANIE PRAWDOPODOBIENSTWA DLA ROZKŁADU *F*-SNEDECORA

STRESZCZENIE

Procedura *PFSnedecor* oblicza prawdopodobieństwo $P(F > F_0)$ dla dowolnych liczb stopni swobody, wykorzystując wzory dokładne.

Dane:

F_0 — wartość funkcji testowej *F*-Snedecora,

n — liczba stopni swobody zmiennej występującej w liczniku,

m — liczba stopni swobody zmiennej występującej w mianowniku.

Przez zastosowanie tej procedury, proces weryfikowania hipotez wykorzystujących rozkład *F*-Snedecora sprowadza się do obliczenia prawdopodobieństwa i ewentualnego porównania go z ustalonym poziomem ufności. Nie ma zatem potrzeby korzystania z tablic rozkładu *F*-Snedecora.